
This handout shows an example of hyperelastic problem using Neo-Hookean 
Hyperelastic material constants. The strain energy potential is given by 

where µ is the initial shear modulus, 

d is material incompressibility, 

1I is the first deviatoric strain invariant, 
 J is the determinant of the elastic deformation gradient. 

 
The Neo-Hookean option (TB,HYPER,,,,NEO) represents the simplest form of strain 
energy potential, and has an applicable strain range of 20-30%.  An example input listing 
showing a typical use of the Neo-Hookean option is presented below. 
 
TB,HYPER,1,,,NEO          !Activate Neo-Hookean data table 
TBDATA,1,0.577148         !Define µ shear modulus 
TBDATA,2,7.0e-5           !Define incompressibility parameter 
                           !(as 2/K, K is the bulk modulus) 
 
Example: Hyperelastic Circular Plate 
 
A flat circular membrane made of a rubber material is subjected to uniform water 
pressure. The edges of the membrane are fixed. Determine the response as pressure is 
increased to 50 psi. 
 

 
Figure 1. Hyperelastic Circular Plate Project Sketch 



Table 1 Material, geometric and loading properties 
Material Properties Geometric Properties Loading 
m = 200 psi R = 7.5 in  

T = 0.5 in 
Pres = 25 psi 

 
The full circular plate is reduced to a 7.5 degree sector for analysis. The mid-plane of the 
outer edge of the circle is considered to be fixed. A pressure of 25 psi is applied to the 
bottom surface of the shell sector. SHELL181, a finite strain layered shell, is used in this 
simulation. Consult ANSYS Element Manual for detailed description of this element. 
 
Table 2 Results comparison 
 Target ANSYS Ratio 
Uz at 4 psi 2.25 2.275 0.989 
Uz at 9.8 psi 3.2 3.379 0.947 
Uz at 15.3 psi 4.1 4.357 0.941 
Uz at 18.8 psi 4.9 5.095 0.962 
Uz at 24 psi 6.2 6.989 0.887 
 
Input Listing 
/PREP7 
ET,1,SHELL181 
R,1,0.5, 
N  ,     1 ,        0.,         0. 
N  ,     2 ,   0.17143,         0. 
N  ,     3 ,   0.47143,         0. 
N  ,     4 ,   0.90000,         0. 
N  ,     5 ,    1.4571,         0. 
N  ,     6 ,    2.1429,         0. 
N  ,     7 ,    2.9571,         0. 
N  ,     8 ,    3.9000,         0. 
N  ,     9 ,    4.9714,         0. 
N  ,    10 ,    6.1714,         0. 
N  ,    11 ,    7.5000,         0. 
N  ,   101 ,        0.,         0. 
N  ,   102 ,   0.16996,    2.23759E-02 
N  ,   103 ,   0.46740,    6.15338E-02 
N  ,   104 ,   0.89230,    0.11747 
N  ,   105 ,    1.4447,    0.19020 
N  ,   106 ,    2.1245,    0.27970 
N  ,   107 ,    2.9318,    0.38598 
N  ,   108 ,    3.8666,    0.50905 
N  ,   109 ,    4.9289,    0.64890 
N  ,   110 ,    6.1186,    0.80553 
N  ,   111 ,    7.4358,    0.97895 
E,    1,       2,     102,     102 
E,    2,       3,     103,     102 
E,    3,       4,     104,     103 
E,    4,       5,     105,     104 
E,    5,       6,     106,     105 
E,    6,       7,     107,     106 
E,    7,       8,     108,     107 
E,    8,       9,     109,     108 
E,    9,      10,     110,     109 
E,   10,      11,     111,     110 
TB,HYPER,1,,,NEO 
TBDATA,1,200,0 
D,  1,UY,0.0,, 11,1  ,ROTX,ROTZ 
D,102,UY,0.0,,111,1  ,ROTX,ROTZ 
D, 11,UX,0.0,,111,100,UY  ,UZ 
D,  1,UX,0.0,,   ,   ,UY  ,ROTX,ROTY,ROTZ 



     
LOCAL, 11, 0, 0.0,0.0,0.0, 7.5,0.0,0.0 
NROTAT,102,111,1 
     
AUTOTS,ON 
NSUBST, 400, 1200,25 
NLGEOM,ON 
NROPT,FULL, ,OFF 
OUTRES, ALL, ALL, 
SF,ALL,PRES,25 
NEQITR,20 
/AUTO,1 
/VIEW,1,,,1 
/ANG,1 
/ESHAPE,1 
EPLOT 
FINISH 
     
/SOLUTION 
SOLVE 
FINISH 
     
/POST1 
/NOPR     !SUPPRESS GRAPHING DATA 
/VIEW,1,,-1 
/ANG,1 
/USER 
/FOCUS,1,4,,8,0    !SET UP CENTER OF GRAPHICS SCREEN FOR DISPLACEMENT PLOT 
/DIST,,12        !SET DISTANCE TO ZOOM OUT 
/TRIAD,OFF 
SET,FIRST       !SET DISPLACEMENT DATA FOR FIRST SUBSTEP 
PLDISP,0      !PLOT DISPLACEMENT DATA 
/NOERASE       !SET DISPLAY TO OVERLAY PLOTS 
SET,,10 
PLDISP,0 
SET,,20 
PLDISP,0 
SET,,25 
PLDISP,0 
SET,LAST 
PLDISP,1       !PLOT FINAL DISPLACEMENT WITH ORIGINAL POSITION 
/ERASE 
/TRIAD,ON 
/GOPR 
/ESHAPE,0 
FINISH 
/POST26 
/XRANGE,0,3.0 
/YRANGE,0,1 
/AXLAB,X,UZ OF CENTER/R-INITIAL 
/AXLAB,Y,THICKNESS/ORIGINAL THICKNESS 
NSOL,2,1,U,Z,UZ_1 
ESOL,3,1, ,SMIS,17,TH_1 
ADD,4,2, , ,UZRATIO, , ,0.13333333,0,0, 
ADD,5,3, , ,SH.181, , ,2,0,0, 
/COLOR,CURVE,MRED 
XVAR,4 
PLVAR,5 
/XRANGE,0,10 
/YRANGE,0,60 
/AXLAB,X,UZ OF CENTER (IN) 
/AXLAB,Y,PRESSURE (LB/SQ IN) 
/COLOR,CURVE,YGRE 
NSOL,2,1,U,Z,UZ_1 
PROD,7,1, , ,SH.181 , , ,25,0,0,  !MULTIPLY SOLUTION BY 25 
/COLOR,CURVE,MRED 
XVAR,2     !SPECIFY X VARIABLE TO BE DISPLAYED 
PLVAR,7    !DISPLAY SOLUTION IN GRPH FILE 
PRVAR,7,2   !LIST VARIABLE 7 VERSUS VARIABLE 2 
FINISH 

